Complex Projective solution set of circle $x^2 + y^2 = 1$

We show that the solution set of the circle $x^2 + y^2 = 1$ is topologically equivalent to the unit sphere. It then follows that the complex projective solution space of any non-singular conic is also a sphere.

Parabola

In fact our first step is to show that the complex projective solution set of the parabola $y = x^2$ is topologically equivalent to the complex projective paraboloid $z = x^2 + y^2$ which we know is topologically a sphere from Chapter 2 of my Surface Story.

We use the simple fact that for any complex number w then y = w^2 gives the solution {w,w^2} for y=x^2. In Mathematica

```
w = RandomComplex[{-10 - 10 I, 10 + 10 I}]
(y - x ^ 2) /. Thread[{x, y} → {w, w ^ 2}]
```

Out[◦]= -3.70333 - 1.25708 ⅈ

Out[◦]= 0. + 0. ⅈ

Conversely given y then the two complex square roots give two solutions for y-x^2.

```
In[ ◦ ]:= w = RandomComplex[{-10 - 10 I, 10 + 10 I}]
u = Sqrt[w]
v = -Sqrt[w]
(y - x ^ 2) /. Thread[{x, y} → {u, w}]
(y - x ^ 2) /. Thread[{x, y} → {v, w}]
```

Out[◦]= 9.03471 - 3.923 ⅈ

Out[◦]= 3.07281 - 0.638339 ⅈ

Out[◦]= -3.07281 + 0.638339 ⅈ

Out[◦]= 0. + 4.44089 × 10^{-16} ⅈ

Out[◦]= 0. + 4.44089 × 10^{-16} ⅈ

Thus using the built in Mathematica functions **ReIm** and **Abs** we can write down continuous mutually inverse functions **Pdown** from the solution space to the unit sphere and **Pup** from the unit sphere back to the solution space.

```
In[ ◦ ]:= Pdown[{u_, v_}] := Append[ReIm[u], Abs[v]]
Pup[{x_, y_, z_}] := {x + I y, (x + I y) ^ 2}
```

Thus for the **u,v,w** immediately above

```
In[ ]:= a = Pdown[{u, w}]
        b = Pdown[{v, w}]
Out[ ]= {3.07281, -0.638339, 9.84966}
Out[ ]= {-3.07281, 0.638339, 9.84966}
```

Note inverses

```
In[ ]:= {u, w}
        Pup[a]
        {v, w}
        Pup[b]
Out[ ]= {3.07281 - 0.638339 i, 9.03471 - 3.923 i}
Out[ ]= {3.07281 - 0.638339 i, 9.03471 - 3.923 i}
Out[ ]= {-3.07281 + 0.638339 i, 9.03471 - 3.923 i}
Out[ ]= {-3.07281 + 0.638339 i, 9.03471 - 3.923 i}
```

More generally

```
In[ ]:= {r, s} = RandomReal[{-10, 10}, 2];
        {r, s, r^2 + s^2}
        c = Pup[{r, s, r^2 + s^2}]
        (y - x^2) /. Thread[{x, y} → c]
        Pdown[c]
Out[ ]= {8.54435, -3.68288, 86.5696}
Out[ ]= {8.54435 - 3.68288 i, 59.4423 - 62.9357 i}
Out[ ]= 0. + 0. i
Out[ ]= {8.54435, -3.68288, 86.5696}
```

illustrating that **Pdown**, **Pup** go to the right places and are inverses for finite solutions of $y - x^2$ and finite points on $z = x^2 + y^2$.

The projective equation for the parabola is $y\,w - x^2 = 0$, setting w=0 give simply $x^2 = 0$ so x=0. Hence the only complex projective infinite point on the parabola is {0,1,0}. Likewise the only projective infinite point on the paraboloid is {0,0,1,0}. Our transformations do not allow for infinite points but we can see we are on the right track by

```
In[ ]:= Pup[{100, 0, 10000}]
Out[ ]= {100, 10000}
```

As a number in projective space this is

```
(100, 10000, 1}
```

which is equivalent by homogeneity to

```
In[ ]:= N[{100, 10 000, 1}/10 000]
```

```
Out[ ]= {0.01, 1., 0.0001}
```

Summarizing, some points in the complex projective space visualized by the paraboloid with their complex coordinates on the right.

```
In[ ]:= Row[{Show[ContourPlot3D [x^2 + y^2 == z, {x, -4, 4}, {y, -4, 4}, {z, 0, 11},
      Mesh → None, ContourStyle → Opacity[.8], Axes → False, Boxed → False],
    ParametricPlot3D [{t, 0, t^2}, {t, -3.3, 3.3}, PlotStyle → Blue],
    Graphics3D [{{Red, Ball[Passoc[#], .2] & /@ Keys[Passoc]},
      {Black, Text["a", {0.3`, -0.3`, 0.4`}], Text["b", {1.3`, -0.3`, 1.4`}],
       Text["c", {2.3`, -0.3`, 4.4`}], Text[Style["d", 14], Passoc["d"] + {.3, -.3, .3}],
       Text["e", {3.372813982557639`, -0.9383394620391801`, 10.249663040198211`}],
       Text["g", {1.3`, 1.7`, 5.4`}], Text["h", {-0.7`, 2.2`, 7.8`}],
       Text["k", {2.4, 2.2, 9.3}]}}], ViewPoint → {4, 2, 0}, ImageSize → 200],
   Column [{Row[{"a = ", ap}], Row[{"b = ", bp}], Row[{"c = ", cp}],
     Row[{"d =", dp}], Row[{"e = ", ep}], Row[{"f = ", fp}],
     Row[{"g = ", gp}], Row[{"h = ", hp}], Row[{"k = ", kp}]}]}]
```

Out[ ]=



a = {0, 0}
b = {1, 1}
c = {2, 4}
d ={2.2 + 1.2 ⅈ, 3.4 + 5.28 ⅈ}
e = {3.07281 - 0.638339 ⅈ, 9.03471 - 3.923 ⅈ}
f = {-3.07281 + 0.638339 ⅈ, 9.03471 - 3.923 ⅈ}
g = {1 + 2 ⅈ, -3 + 4 ⅈ}
h = {-1. + 2.5 ⅈ, -5.25 - 5. ⅈ}
k = {2.5 + 1.7 ⅈ, 3.36 + 8.5 ⅈ}

Note that the blue curve is the real parabola.

Circle

This follows immediately since the circle is projectively equivalent to the parabola. However we give explicit transformations for the convenience of the reader.

We recall from my *Plane Curve Book* that there is an invertible projective linear transformation, that is FLT, from the parabola to the circle with matrix

```
In[ ]:= P2C = {{1, 0, 0}, {0, -0.5`, 0.5`}, {0, -0.5`, -0.5`}}
```

```
Out[ ]= {{1, 0, 0}, {0, -0.5, 0.5}, {0, -0.5, -0.5}}
```

and, from my *Surface Story* an invertible projective linear transformation from the paraboloid to the sphere given by matrix

In[ ]:=
```
PB2SP = {{0, 0, 1/2, -1/2}, {1, 0, 0, 0}, {0, 1, 0, 0}, {0, 0, 1/2, 1/2}}
```

Out[ ]= $\left\{\left\{0, 0, \frac{1}{2}, -\frac{1}{2}\right\}, \{1, 0, 0, 0\}, \{0, 1, 0, 0\}, \left\{0, 0, \frac{1}{2}, \frac{1}{2}\right\}\right\}$

We can then write down our transformations

In[ ]:=
```
Cdown[{x_, y_}] :=
 TransformationFunction [PB2SP][Pdown[TransformationFunction [Inverse[P2C]][{-y, x}]]]
Cup[{x_, y_, z_}] := TransformationFunction [P2C][
  Pup[TransformationFunction [Inverse[PB2SP]][{y, -x, z}]]]
```

The unexpected coordinates are because of a slight inconsistency of the two previously given transfor - mations .

In[ ]:=
```
Clear[a, b, c]
```

To illustrate

In[ ]:=
```
a = N[{Cos[Pi / 3], Sin[Pi / 3]}]
ad = N[Cdown[{Cos[Pi / 3], Sin[Pi / 3] }]]
```

Out[ ]= {0.5, 0.866025}

Out[ ]= {0.5, 0.866025, 0.}

This suggests the fact that **Cdown** embeds the circle as the equator of the unit sphere. In fact it is easy to construct complex solutions just by using complex arguments for **Cos** and **Sin.**

In[ ]:=
```
b = N[{Cos[.2 - .3 ⅈ], Sin[.2 - .3 ⅈ]}]
```

Out[ ]= {1.0245 + 0.0604988 ⅈ, 0.207677 − 0.29845 ⅈ}

In[ ]:=
```
Simplify [(x ^ 2 + y ^ 2 - 1) /. Thread [{x, y} → b]]
```

Out[ ]= $-2.22045 \times 10^{-16} - 1.38778 \times 10^{-17}\, i$

In[ ]:=
```
bd = Cdown[b]
(x ^ 2 + y ^ 2 + z ^ 2 - 1) /. Thread [{x, y, z} → bd]
```

Out[ ]= {0.937559, 0.190053, 0.291313}

Out[ ]= 1.11022×10^{-16}

One observation is that although the real circle is bounded, the complex circle is not. For example let

```
In[ ]:= c = {15 ⅈ, Sqrt[1 + 15 ^ 2]}
        N[Norm[c]]
        (x ^ 2 + y ^ 2 - 1) /. Thread[{x, y} → c]
```

Out[ ]= $\{15\, i,\ \sqrt{226}\}$

Out[ ]= 21.2368

Out[ ]= 0

But **Cdown[c]** is still on the unit sphere, but close to the north pole

```
In[ ]:= cd = Cdown[c]
Out[ ]= {0., 0.066519, 0.997785}
```

In fact, using homogenous projective coordinates $x^2 + y^2 = w^2$ for the circle the infinite part of the complex unit circle is found by setting $w = 0$ and noticing

```
In[ ]:= Expand[(x + ⅈ y) (x - ⅈ y)]
```

Out[ ]= $x^2 + y^2$

So $x + i\, y = 0$ or $x - i\, y = 0$. By homogeneity we can divide by x or y getting the two homogeneous solutions $\{i, 1, 0\}$ or $(-i, 1, 0\}$. Note that

```
In[ ]:= (x ^ 2 + y ^ 2 - 1) /. Thread[{x, y} → {t ⅈ, Sqrt[1 - (t ⅈ) ^ 2]}]
Out[ ]= 0
```

so gives point in our solution set for arbitrary large t, t positive or negative

```
In[ ]:= Limit[Cdown[{t ⅈ, Sqrt[1 - (t ⅈ) ^ 2]}], t → ∞]
        Limit[Cdown[{t ⅈ, Sqrt[1 - (t ⅈ) ^ 2]}], t → -∞]
Out[ ]= {0., 0., 1.}
Out[ ]= {0., 0., -1.}
```

and infer that $\{i, 1\}$ goes to the north pole and $\{1, i\}$ goes to the south pole of the sphere. Our picture here is

```
In[ ]:= a = {0.5`, 0.8660254037844386`};
        b = {1.0245013402279206` + 0.06049884291265947` ⅈ,
            0.20767670305628436` - 0.29845016188195167` ⅈ};
        c = {0.` + 15.` ⅈ, 15.033296378372908`};
        d = {0.49999999999999956` + 1.1618950038622249` ⅈ, 1.5` - 0.38729833462074126` ⅈ};
        e = {0.8988764044943819` - 0.1863272354132246` ⅈ,
            0.5617977528089886` + 0.2981235766611593` ⅈ};
```

In[ ]:=
```
ad = Cdown[a]
bd = Cdown[b]
cd = Cdown[c]
dd = Cdown[d]
ed = Cdown[e]
```

Out[ ]= {0.5, 0.866025, 0.}

Out[ ]= {0.937559, 0.190053, 0.291313}

Out[ ]= {0., 0.066519, 0.997785}

Out[ ]= {0.2, 0.6, 0.774597}

Out[ ]= {0.8, 0.5, -0.331662}

In[ ]:=
```
Row[{Show[ContourPlot3D [x^2 + y^2 + z^2 == 1, {x, -1.01, 1.01}, {y, -1.01, 1.01},
    {z, -1.01, 1.01}, Mesh → None, ContourStyle → Opacity[.8], Axes → False,
    Boxed → False], ParametricPlot3D [{Cos[t], Sin[t], 0}, {t, -Pi, Pi}, PlotStyle → Blue],
   Graphics3D [{{White, Ball[{0, 0, 1}, .05], Ball[{0, 0, -1}, .05]},
     {Red, Ball[ad, .05], Ball[bd, .05], Ball[cd, .05], Ball[dd, .05], Ball[ed, .05]},
     {Black, Text["a", ad + {0.1`, -0.1`, 0.1`}], Text["b", bd + {0`, -0.1`, 0.1`}],
      Text["c", cd + {-0.05`, 0.1`, -0.05`}], Text[Style["d", 14], dd + {.15, .2, .2}], Text[
       Style["e", 14], ed + {.04, .03, .22}], Text["i", {0, 0, 1} + {0`, -0.1`, -0.03`}]}}],
   ViewPoint → {2, 1, 1.5}, ImageSize → 200], Column[{Row[{"a = ", a}], Row[{"b = ", b}],
    Row[{"c = ", c}], Row[{"d =", d}], Row[{"e = ", e}], "i    (infinite  point)" }]}]
```

Out[ ]=



a = {0.5, 0.866025}
b = {1.0245 + 0.0604988 ⅈ, 0.207677 - 0.29845 ⅈ}
c = {0. + 15. ⅈ, 15.0333}
d ={0.5 + 1.1619 ⅈ, 1.5 - 0.387298 ⅈ}
e = {0.898876 - 0.186327 ⅈ, 0.561798 + 0.298124 ⅈ}
i   (infinite point)

In[ ]:=
```
d = Cup[dd]
```

Out[ ]= {0.5 + 1.1619 ⅈ, 1.5 - 0.387298 ⅈ}